


Parameter Estimation for Photovoltaic Diodes



The behavior of a photovoltaic diode is described by this equation

$$pv := I_f = I_{ph} - I_{SD} \cdot \left(\exp\left(\frac{(V_f + I_f \cdot R_s)}{n \cdot V_t}\right) - 1 \right) - \frac{(V_f + I_f \cdot R_s)}{R_{sh}}$$

where

- I_f and V_f are the experimental current and voltage
- I_{sh} and R_{sh} are the shunt current and resistance
- R_s is the series resistance
- I_{SD} is the diode saturation current (normal diode current)
- I_{ph} is the cell-generated photocurrent
- $V_t = k T / q$ is the thermal voltage (where k is the Boltzmann - constant, T is the ambient temperature and q is the electron charge)
- n is the ideality factor

This application will rearrange this equation to give I_f in terms of the LambertW equation, and find the best-fit parameters (I_{ph} , I_{SD} , n , R_s , R_{sh}) against experimental values of I_f and V_f

Rearrange photovoltaic diode equation for forward current

$$I_r := \text{solve}(pv, I_f)$$


$$I_r = -\frac{-\text{LambertW}\left(\frac{\frac{R_{sh} \cdot (I_{SD} \cdot R_s + I_{ph} \cdot R_s + V_f)}{n \cdot V_t \cdot (R_s + R_{sh})}}{-\frac{I_{SD} \cdot R_s \cdot R_{sh} \cdot e^{\frac{R_{sh} \cdot (I_{SD} \cdot R_s + I_{ph} \cdot R_s + V_f)}{n \cdot V_t \cdot (R_s + R_{sh})}}}{-R_s \cdot V_t \cdot n - R_{sh} \cdot V_t \cdot n}}\right) + \frac{R_{sh} \cdot (I_{SD} \cdot R_s + I_{ph} \cdot R_s + V_f)}{n \cdot V_t \cdot (R_s + R_{sh})}}{R_s} \cdot n \cdot V_t + V_f$$

Import experimental data

```
data_file_path := FileTools:-JoinPath([kernelopts(mapledir), "toolbox", "MapleFlow", "data", "examples", "diodeexperimentaldata.xlsx"])
```

```
data := ImportMatrix(data_file_path)
```

```
plot(data, style = point) =
```



Define parameters

$$T := 306$$

$$k := 1.380650 \times 10^{-23}$$

$$q := 1.602176 \times 10^{-19}$$

$$V_t := \frac{k \cdot T}{q} = 0.026$$

Unknown values to be found

```
indets(I_r, name) = {I_SD, I_ph, R_s, R_sh, V_f, n}
```

Fit parameters to data

```
res := Statistics:-NonlinearFit(I_r, data, V_f, parameterranges = [I_ph = 0.1..1, I_SD = 0..0.0001, n = 1..2, R_s = 0.01..0.1, R_sh = 1..100], output = parametervalues, iterationlimit = 50, optimalitytolerance = 0.01)
```

```
res = [I_SD = 9.325 × 10⁻⁶, I_ph = 0.766, R_s = 0.016, R_sh = 51.329, n = 1.924]
```

Plot the experimental data against the calibrated model

```
p1 := plot(data, style = point, legend = "Experimental Data")
```

```
p2 := plot(eval(I_r, res), V_f = min(data[ .., 1]) .. max(data[ .., 1]), legend = "Model Curve")
```

```
plots:-display(p1, p2) =
```

